Rules for integrands involving inverse hyperbolic sines

1. $\int u\,(a+b\,\text{ArcSinh}[c+d\,x])^n\,dx$

1: $\int (a+b\,\text{ArcSinh}[c+d\,x])^n\,dx$

Derivation: Integration by substitution

Rule:

$$\int (a+b\,\text{ArcSinh}[c+d\,x])^n\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int (a+b\,\text{ArcSinh}[x])^n\,dx,\ x,\ c+d\,x\right]$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcSinh[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,n},x]
```

2: $\int (e+f\,x)^m\,(a+b\,\text{ArcSinh}[c+d\,x])^n\,dx$

Derivation: Integration by substitution

Rule:

$$\int (e+f\,x)^m\,(a+b\,\text{ArcSinh}[c+d\,x])^n\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int \left(\frac{d\,e-c\,f}{d}+\frac{f\,x}{d}\right)^m (a+b\,\text{ArcSinh}[x])^n\,dx,\ x,\ c+d\,x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSinh[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcSinh[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

3: $\int (A + B x + C x^2)^p (a + b \operatorname{ArcSinh}[c + d x])^n \, dx$ when $B (1 + c^2) - 2 A c d == 0 \land 2 c C - B d == 0$

Derivation: Integration by substitution

Basis: If $B (1 + c^2) - 2 A c d == 0 \land 2 c C - B d == 0$, then $A + B x + C x^2 == \frac{C}{d^2} + \frac{C}{d^2} (c + d x)^2$

Rule: If $B (1 + c^2) - 2 A c d == 0 \land 2 c C - B d == 0$, then

$$\int (A + B x + C x^2)^p (a + b \operatorname{ArcSinh}[c + d x])^n \, dx \to \frac{1}{d} \operatorname{Subst}\left[\int \left(\frac{C}{d^2} + \frac{C x^2}{d^2}\right)^p (a + b \operatorname{ArcSinh}[x])^n \, dx, x, c + d x\right]$$

Program code:

```
Int[(A_.+B_.*x_+C_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[(C/d^2+C/d^2*x^2)^p*(a+b*ArcSinh[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,A,B,C,n,p},x] && EqQ[B*(1+c^2)-2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

4: $\int (e+f x)^m (A+B x+C x^2)^p (a+b \text{ArcSinh}[c+d x])^n \, dx$ when $B(1+c^2)-2 A c d == 0 \land 2 c C - B d == 0$

Derivation: Integration by substitution

Basis: If $B(1+c^2)-2 A c d == 0 \land 2 c C - B d == 0$, then $A+B x+C x^2 == \frac{C}{d^2}+\frac{C}{d^2}(c+d x)^2$

Rule: If $B(1+c^2)-2 A c d == 0 \land 2 c C - B d == 0$, then

$$\int (e+f x)^m (A+B x+C x^2)^p (a+b \text{ArcSinh}[c+d x])^n \, dx \to \frac{1}{d} \text{Subst}\left[\int \left(\frac{d e-c f}{d}+\frac{f x}{d}\right)^m \left(\frac{C}{d^2}+\frac{C x^2}{d^2}\right)^p (a+b \text{ArcSinh}[x])^n \, dx,\ x,\ c+d x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(A_.+B_.*x_+C_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(C/d^2+C/d^2*x^2)^p*(a+b*ArcSinh[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n,p},x] && EqQ[B*(1+c^2)-2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

2. $\int (a+b \text{ArcSinh}[c+d x^2])^n \, dx$ when $c^2 == -1$

1. $\int (a+b \text{ArcSinh}[c+d x^2])^n \, dx$ when $c^2 == -1 \land n > 0$

1: $\int \sqrt{a+b \text{ArcSinh}[c+d x^2]} \, dx$ when $c^2 == -1$

Derivation: Integration by parts

Note: This antiderivative is probably better expressed in terms of error functions...

Rule: If $c^2 == -1$, then

$$\int \sqrt{a+b \text{ArcSinh}[c+d x^2]} \, dx \to x \sqrt{a+b \text{ArcSinh}[c+d x^2]} - b d \int \frac{x^2}{\sqrt{2 c d x^2+d^2 x^4} \sqrt{a+b \text{ArcSinh}[c+d x^2]}} \, dx$$

$$\longrightarrow x\sqrt{a+b\,\text{ArcSinh}[c+d\,x^2]}\,-$$
$$\frac{\sqrt{\pi}\,x\left(\text{Cosh}\left[\frac{a}{2b}\right]-c\,\text{Sinh}\left[\frac{a}{2b}\right]\right)\text{FresnelC}\left[\sqrt{-\frac{c}{\pi b}}\,\sqrt{a+b\,\text{ArcSinh}[c+d\,x^2]}\right]}{\sqrt{-\frac{c}{b}}\left(\text{Cosh}\left[\frac{1}{2}\text{ArcSinh}[c+d\,x^2]\right]+c\,\text{Sinh}\left[\frac{1}{2}\text{ArcSinh}[c+d\,x^2]\right]\right)}\,+$$
$$\frac{\sqrt{\pi}\,x\left(\text{Cosh}\left[\frac{a}{2b}\right]+c\,\text{Sinh}\left[\frac{a}{2b}\right]\right)\text{FresnelS}\left[\sqrt{-\frac{c}{\pi b}}\,\sqrt{a+b\,\text{ArcSinh}[c+d\,x^2]}\right]}{\sqrt{-\frac{c}{b}}\left(\text{Cosh}\left[\frac{1}{2}\text{ArcSinh}[c+d\,x^2]\right]+c\,\text{Sinh}\left[\frac{1}{2}\text{ArcSinh}[c+d\,x^2]\right]\right)}$$

Program code:

```
Int[Sqrt[a_.+b_.*ArcSinh[c_+d_.*x_^2]],x_Symbol] :=
  x*Sqrt[a+b*ArcSinh[c+d*x^2]] -
  Sqrt[Pi]*x*(Cosh[a/(2*b)]-c*Sinh[a/(2*b)])*FresnelC[Sqrt[-c/(Pi*b)]*Sqrt[a+b*ArcSinh[c+d*x^2]]]/
    (Sqrt[-(c/b)]*(Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[ArcSinh[c+d*x^2]/2])) +
  Sqrt[Pi]*x*(Cosh[a/(2*b)]+c*Sinh[a/(2*b)])*FresnelS[Sqrt[-c/(Pi*b)]*Sqrt[a+b*ArcSinh[c+d*x^2]]]/
    (Sqrt[-(c/b)]*(Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[ArcSinh[c+d*x^2]/2])) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,-1]
```

2: $\int\left(a+b\,\text{ArcSinh}[c+d\,x^2]\right)^n dx$ when $c^2==-1\land n>1$

Derivation: Integration by parts twice

Basis: If $c^2==-1$, then $\partial_x\left(a+b\,\text{ArcSinh}[c+d\,x^2]\right)^n==\frac{2\,b\,d\,n\,x\left(a+b\,\text{ArcSinh}[c+d\,x^2]\right)^{n-1}}{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}$

Basis: $\frac{x^2}{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}==\partial_x\frac{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}{d^2\,x}$

Rule: If $c^2==-1\land n>1$, then

$$\int\left(a+b\,\text{ArcSinh}[c+d\,x^2]\right)^n dx\longrightarrow x\left(a+b\,\text{ArcSinh}[c+d\,x^2]\right)^n-2\,b\,d\,n\int\frac{x^2\left(a+b\,\text{ArcSinh}[c+d\,x^2]\right)^{n-1}}{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}\,dx$$

$$\to x\,(a+b\,\text{ArcSinh}[c+d\,x^2])^n-\frac{2\,b\,n\,\sqrt{2\,c\,d\,x^2+d^2\,x^4}\,(a+b\,\text{ArcSinh}[c+d\,x^2])^{n-1}}{d\,x}+4\,b^2\,n\,(n-1)\int(a+b\,\text{ArcSinh}[c+d\,x^2])^{n-2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_+d_.*x_^2])^n_,x_Symbol] :=
  x*(a+b*ArcSinh[c+d*x^2])^n -
  2*b*n*Sqrt[2*c*d*x^2+d^2*x^4]*(a+b*ArcSinh[c+d*x^2])^(n-1)/(d*x) +
  4*b^2*n*(n-1)*Int[(a+b*ArcSinh[c+d*x^2])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,-1] && GtQ[n,1]
```

2. $\int(a+b\,\text{ArcSinh}[c+d\,x^2])^n\,dx$ when $c^2==-1\land n<0$

1: $\int\frac{1}{a+b\,\text{ArcSinh}[c+d\,x^2]}\,dx$ when $c^2==-1$

Rule: If $c^2==-1$, then

$$\int\frac{1}{a+b\,\text{ArcSinh}[c+d\,x^2]}\,dx\to$$

$$\frac{x\left(c\,\text{Cosh}\left[\frac{a}{2\,b}\right]-\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{CoshIntegral}\left[\frac{1}{2\,b}\,(a+b\,\text{ArcSinh}[c+d\,x^2])\right]}{2\,b\left(\text{Cosh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]+c\,\text{Sinh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]\right)}+\frac{x\left(\text{Cosh}\left[\frac{a}{2\,b}\right]-c\,\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{SinhIntegral}\left[\frac{1}{2\,b}\,(a+b\,\text{ArcSinh}[c+d\,x^2])\right]}{2\,b\left(\text{Cosh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]+c\,\text{Sinh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]\right)}$$

Program code:

```
Int[1/(a_.+b_.*ArcSinh[c_+d_.*x_^2]),x_Symbol] :=
  x*(c*Cosh[a/(2*b)]-Sinh[a/(2*b)])*CoshIntegral[(a+b*ArcSinh[c+d*x^2])/(2*b)]/
    (2*b*(Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[(1/2)*ArcSinh[c+d*x^2]])) +
  x*(Cosh[a/(2*b)]-c*Sinh[a/(2*b)])*SinhIntegral[(a+b*ArcSinh[c+d*x^2])/(2*b)]/
    (2*b*(Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[(1/2)*ArcSinh[c+d*x^2]])) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,-1]
```

2: $\int \frac{1}{\sqrt{a + b\,\text{ArcSinh}[c + d\,x^2]}}\,dx$ when $c^2 == -1$

Rule: If $c^2 == -1$, then

$$\int \frac{1}{\sqrt{a + b\,\text{ArcSinh}[c + d\,x^2]}}\,dx \to$$

$$\frac{(c+1)\sqrt{\frac{\pi}{2}}\,x\left(\text{Cosh}\left[\frac{a}{2b}\right] - \text{Sinh}\left[\frac{a}{2b}\right]\right)\text{Erfi}\left[\frac{1}{\sqrt{2b}}\sqrt{a + b\,\text{ArcSinh}[c + d\,x^2]}\right]}{2\sqrt{b}\left(\text{Cosh}\left[\frac{1}{2}\text{ArcSinh}[c + d\,x^2]\right] + c\,\text{Sinh}\left[\frac{1}{2}\text{ArcSinh}[c + d\,x^2]\right]\right)} + \frac{(c-1)\sqrt{\frac{\pi}{2}}\,x\left(\text{Cosh}\left[\frac{a}{2b}\right] + \text{Sinh}\left[\frac{a}{2b}\right]\right)\text{Erf}\left[\frac{1}{\sqrt{2b}}\sqrt{a + b\,\text{ArcSinh}[c + d\,x^2]}\right]}{2\sqrt{b}\left(\text{Cosh}\left[\frac{1}{2}\text{ArcSinh}[c + d\,x^2]\right] + c\,\text{Sinh}\left[\frac{1}{2}\text{ArcSinh}[c + d\,x^2]\right]\right)}$$

Program code:

```
Int[1/Sqrt[a_.+b_.*ArcSinh[c_+d_.*x_^2]],x_Symbol] :=
  (c+1)*Sqrt[Pi/2]*x*(Cosh[a/(2*b)]-Sinh[a/(2*b)])*Erfi[Sqrt[a+b*ArcSinh[c+d*x^2]]/Sqrt[2*b]]/
    (2*Sqrt[b]*(Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[ArcSinh[c+d*x^2]/2])) +
  (c-1)*Sqrt[Pi/2]*x*(Cosh[a/(2*b)]+Sinh[a/(2*b)])*Erf[Sqrt[a+b*ArcSinh[c+d*x^2]]/Sqrt[2*b]]/
    (2*Sqrt[b]*(Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[ArcSinh[c+d*x^2]/2])) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,-1]
```

3. $\int \left(a + b\,\text{ArcSinh}[c + d\,x^2]\right)^n dx$ when $c^2 == -1 \land n < -1$

1: $\int \frac{1}{\left(a + b\,\text{ArcSinh}[c + d\,x^2]\right)^{3/2}}\,dx$ when $c^2 == -1$

Derivation: Integration by parts

Basis: If $c^2 == -1$, then $-\frac{b\,d\,x}{\sqrt{2\,c\,d\,x^2 + d^2\,x^4}\,\left(a + b\,\text{ArcSinh}[c + d\,x^2]\right)^{3/2}} == \partial_x \frac{1}{\sqrt{a + b\,\text{ArcSinh}[c + d\,x^2]}}$

Rule: If $c^2 == -1$, then

$$\int \frac{1}{\left(a + b\,\text{ArcSinh}[c + d\,x^2]\right)^{3/2}}\,dx \to -\frac{\sqrt{2\,c\,d\,x^2 + d^2\,x^4}}{b\,d\,x\,\sqrt{a + b\,\text{ArcSinh}[c + d\,x^2]}} + \frac{d}{b}\int \frac{x^2}{\sqrt{2\,c\,d\,x^2 + d^2\,x^4}\,\sqrt{a + b\,\text{ArcSinh}[c + d\,x^2]}}\,dx$$

$$\to -\frac{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}{b\,d\,x\,\sqrt{a+b\,\text{ArcSinh}[c+d\,x^2]}} - \left(\left(-\frac{c}{b}\right)^{3/2}\sqrt{\pi}\;x\left(\text{Cosh}\left[\frac{a}{2\,b}\right]-c\,\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{FresnelC}\left[\sqrt{-\frac{c}{\pi\,b}}\;\sqrt{a+b\,\text{ArcSinh}[c+d\,x^2]}\right]\right)\Big/\left(\text{Cosh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]+c\,\text{Sinh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]\right) + \left(\left(-\frac{c}{b}\right)^{3/2}\sqrt{\pi}\;x\left(\text{Cosh}\left[\frac{a}{2\,b}\right]+c\,\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{FresnelS}\left[\sqrt{-\frac{c}{\pi\,b}}\;\sqrt{a+b\,\text{ArcSinh}[c+d\,x^2]}\right]\right)\Big/\left(\text{Cosh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]+c\,\text{Sinh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]\right)$$

Program code:

```
Int[1/(a_.+b_.*ArcSinh[c_+d_.*x_^2])^(3/2),x_Symbol] :=
  -Sqrt[2*c*d*x^2+d^2*x^4]/(b*d*x*Sqrt[a+b*ArcSinh[c+d*x^2]]) -
  (-c/b)^(3/2)*Sqrt[Pi]*x*(Cosh[a/(2*b)]-c*Sinh[a/(2*b)])*FresnelC[Sqrt[-c/(Pi*b)]*Sqrt[a+b*ArcSinh[c+d*x^2]]]/
    (Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[ArcSinh[c+d*x^2]/2]) +
  (-c/b)^(3/2)*Sqrt[Pi]*x*(Cosh[a/(2*b)]+c*Sinh[a/(2*b)])*FresnelS[Sqrt[-c/(Pi*b)]*Sqrt[a+b*ArcSinh[c+d*x^2]]]/
    (Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[ArcSinh[c+d*x^2]/2]) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,-1]
```

2: $\int \frac{1}{(a+b\,\text{ArcSinh}[c+d\,x^2])^2}\,dx$ when $c^2 == -1$

Derivation: Integration by parts

Basis: If $c^2 == -1$, then $-\frac{2\,b\,d\,x}{\sqrt{2\,c\,d\,x^2+d^2\,x^4}\;(a+b\,\text{ArcSinh}[c+d\,x^2])^2} == \partial_x \frac{1}{a+b\,\text{ArcSinh}[c+d\,x^2]}$

Rule: If $c^2 == -1$, then

$$\int \frac{1}{(a+b\,\text{ArcSinh}[c+d\,x^2])^2}\,dx \to -\frac{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}{2\,b\,d\,x\,(a+b\,\text{ArcSinh}[c+d\,x^2])} + \frac{d}{2\,b}\int \frac{x^2}{\sqrt{2\,c\,d\,x^2+d^2\,x^4}\;(a+b\,\text{ArcSinh}[c+d\,x^2])}\,dx$$

$$\to -\frac{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}{2\,b\,d\,x\,(a+b\,\text{ArcSinh}[c+d\,x^2])} + \frac{x\left(\text{Cosh}\left[\frac{a}{2\,b}\right]-c\,\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{CoshIntegral}\left[\frac{1}{2\,b}\,(a+b\,\text{ArcSinh}[c+d\,x^2])\right]}{4\,b^2\left(\text{Cosh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]+c\,\text{Sinh}\left[\frac{1}{2}\,\text{ArcSinh}[c+d\,x^2]\right]\right)} +$$

$$\frac{x\left(c\operatorname{Cosh}\left[\frac{a}{2b}\right]-\operatorname{Sinh}\left[\frac{a}{2b}\right]\right)\operatorname{SinhIntegral}\left[\frac{1}{2b}\left(a+b\operatorname{ArcSinh}\left[c+d\,x^2\right]\right)\right]}{4\,b^2\left(\operatorname{Cosh}\left[\frac{1}{2}\operatorname{ArcSinh}\left[c+d\,x^2\right]\right]+c\operatorname{Sinh}\left[\frac{1}{2}\operatorname{ArcSinh}\left[c+d\,x^2\right]\right]\right)}$$

– Program code:

```
Int[1/(a_.+b_.*ArcSinh[c_+d_.*x_^2])^2,x_Symbol] :=
  -Sqrt[2*c*d*x^2+d^2*x^4]/(2*b*d*x*(a+b*ArcSinh[c+d*x^2])) +
  x*(Cosh[a/(2*b)]-c*Sinh[a/(2*b)])*CoshIntegral[(a+b*ArcSinh[c+d*x^2])/(2*b)]/
    (4*b^2*(Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[ArcSinh[c+d*x^2]/2])) +
  x*(c*Cosh[a/(2*b)]-Sinh[a/(2*b)])*SinhIntegral[(a+b*ArcSinh[c+d*x^2])/(2*b)]/
    (4*b^2*(Cosh[ArcSinh[c+d*x^2]/2]+c*Sinh[ArcSinh[c+d*x^2]/2])) /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,-1]
```

3: $\int\left(a+b\operatorname{ArcSinh}\left[c+d\,x^2\right]\right)^n dx$ when $c^2 == -1 \land n < -1 \land n \neq -2$

Derivation: Inverted integration by parts twice

– Rule: If $c^2 == -1 \land n < -1 \land n \neq -2$, then

$$\int\left(a+b\operatorname{ArcSinh}\left[c+d\,x^2\right]\right)^n dx \longrightarrow$$

$$-\frac{x\left(a+b\operatorname{ArcSinh}\left[c+d\,x^2\right]\right)^{n+2}}{4\,b^2\,(n+1)\,(n+2)}+\frac{\sqrt{2\,c\,d\,x^2+d^2\,x^4}\left(a+b\operatorname{ArcSinh}\left[c+d\,x^2\right]\right)^{n+1}}{2\,b\,d\,(n+1)\,x}+\frac{1}{4\,b^2\,(n+1)\,(n+2)}\int\left(a+b\operatorname{ArcSinh}\left[c+d\,x^2\right]\right)^{n+2}dx$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_+d_.*x_^2])^n_,x_Symbol] :=
  -x*(a+b*ArcSinh[c+d*x^2])^(n+2)/(4*b^2*(n+1)*(n+2)) +
  Sqrt[2*c*d*x^2+d^2*x^4]*(a+b*ArcSinh[c+d*x^2])^(n+1)/(2*b*d*(n+1)*x) +
  1/(4*b^2*(n+1)*(n+2))*Int[(a+b*ArcSinh[c+d*x^2])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,-1] && LtQ[n,-1] && NeQ[n,-2]
```

3: $\int \frac{\text{ArcSinh}[a\,x^p]^n}{x}\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $\frac{\text{ArcSinh}[a\,x^p]^n}{x} == \frac{1}{p}\,\text{Subst}[x^n\,\text{Coth}[x],\,x,\,\text{ArcSinh}[a\,x^p]]\;\partial_x\,\text{ArcSinh}[a\,x^p]$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \frac{\text{ArcSinh}[a\,x^p]^n}{x}\,dx \longrightarrow \frac{1}{p}\,\text{Subst}\left[\int x^n\,\text{Coth}[x]\,dx,\,x,\,\text{ArcSinh}[a\,x^p]\right]$$

Program code:

```
Int[ArcSinh[a_.*x_^p_]^n_./x_,x_Symbol] :=
  1/p*Subst[Int[x^n*Coth[x],x],x,ArcSinh[a*x^p]] /;
FreeQ[{a,p},x] && IGtQ[n,0]
```

4: $\int u\,\text{ArcSinh}\left[\frac{c}{a+b\,x^n}\right]^m\,dx$

Derivation: Algebraic simplification

Basis: $\text{ArcSinh}[z] == \text{ArcCsch}\left[\frac{1}{z}\right]$

Rule:

$$\int u\,\text{ArcSinh}\left[\frac{c}{a+b\,x^n}\right]^m\,dx \longrightarrow \int u\,\text{ArcCsch}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m\,dx$$

Program code:

```
Int[u_.*ArcSinh[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCsch[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

5: $\int \frac{\text{ArcSinh}\left[\sqrt{-1+b x^2}\right]^n}{\sqrt{-1+b x^2}}\, dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\sqrt{b x^2}}{x} == 0$

Basis: $\frac{x\,\text{ArcSinh}\left[\sqrt{-1+b x^2}\right]^n}{\sqrt{b x^2}\,\sqrt{-1+b x^2}} == \frac{1}{b}\,\text{Subst}\left[\frac{\text{ArcSinh}[x]^n}{\sqrt{1+x^2}},\, x,\, \sqrt{-1+b x^2}\right] \partial_x \sqrt{-1+b x^2}$

Rule:

$$\int \frac{\text{ArcSinh}\left[\sqrt{-1+b x^2}\right]^n}{\sqrt{-1+b x^2}}\, dx \to \frac{\sqrt{b x^2}}{x} \int \frac{x\,\text{ArcSinh}\left[\sqrt{-1+b x^2}\right]^n}{\sqrt{b x^2}\,\sqrt{-1+b x^2}}\, dx$$

$$\to \frac{\sqrt{b x^2}}{b x}\,\text{Subst}\left[\int \frac{\text{ArcSinh}[x]^n}{\sqrt{1+x^2}}\, dx,\, x,\, \sqrt{-1+b x^2}\right]$$

Program code:

```
Int[ArcSinh[Sqrt[-1+b_.*x_^2]]^n_./Sqrt[-1+b_.*x_^2],x_Symbol] :=
  Sqrt[b*x^2]/(b*x)*Subst[Int[ArcSinh[x]^n/Sqrt[1+x^2],x],x,Sqrt[-1+b*x^2]] /;
FreeQ[{b,n},x]
```

6. $\int u\, f^{c\,\text{ArcSinh}[a+b\,x]^n}\,dx$ when $n \in \mathbb{Z}^+$

1: $\int f^{c\,\text{ArcSinh}[a+b\,x]^n}\,dx$ when $n \in \mathbb{Z}^+$

- Derivation: Integration by substitution

- Basis: $F[\text{ArcSinh}[a+b\,x]] == \frac{1}{b}\,\text{Subst}[F[x]\,\text{Cosh}[x],\,x,\,\text{ArcSinh}[a+b\,x]]\,\partial_x\,\text{ArcSinh}[a+b\,x]$

- Rule: If $n \in \mathbb{Z}^+$, then

$$\int f^{c\,\text{ArcSinh}[a+b\,x]^n}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int f^{c\,x^n}\,\text{Cosh}[x]\,dx,\,x,\,\text{ArcSinh}[a+b\,x]\right]$$

Program code:

```
Int[f_^(c_.*ArcSinh[a_.+b_.*x_]^n_.),x_Symbol] :=
  1/b*Subst[Int[f^(c*x^n)*Cosh[x],x],x,ArcSinh[a+b*x]] /;
FreeQ[{a,b,c,f},x] && IGtQ[n,0]
```

2: $\int x^m f^{c \operatorname{ArcSinh}[a+b x]^n} \, dx$ when $(m \mid n) \in \mathbb{Z}^+$

- Derivation: Integration by substitution

- Basis: $F[x, \operatorname{ArcSinh}[a+b x]] = \frac{1}{b} \operatorname{Subst}\left[F\left[-\frac{a}{b}+\frac{\operatorname{Sinh}[x]}{b}, x\right] \operatorname{Cosh}[x], x, \operatorname{ArcSinh}[a+b x]\right] \partial_x \operatorname{ArcSinh}[a+b x]$

- Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int x^m f^{c \operatorname{ArcSinh}[a+b x]^n} \, dx \longrightarrow \frac{1}{b} \operatorname{Subst}\left[\int \left(-\frac{a}{b}+\frac{\operatorname{Sinh}[x]}{b}\right)^m f^{c x^n} \operatorname{Cosh}[x] \, dx, x, \operatorname{ArcSinh}[a+b x]\right]$$

- Program code:

```
Int[x_^m_.*f_^(c_.*ArcSinh[a_.+b_.*x_]^n_.),x_Symbol] :=
  1/b*Subst[Int[(-a/b+Sinh[x]/b)^m*f^(c*x^n)*Cosh[x],x],x,ArcSinh[a+b*x]] /;
FreeQ[{a,b,c,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

7. $\int v\,(a+b\,\text{ArcSinh}[u])\,dx$ when u is free of inverse functions

1: $\int \text{ArcSinh}[u]\,dx$ when u is free of inverse functions

Derivation: Integration by parts

Rule: If u is free of inverse functions, then

$$\int \text{ArcSinh}[u]\,dx \longrightarrow x\,\text{ArcSinh}[u]-\int \frac{x\,\partial_x u}{\sqrt{1+u^2}}\,dx$$

Program code:

```
Int[ArcSinh[u_],x_Symbol] :=
  x*ArcSinh[u] -
  Int[SimplifyIntegrand[x*D[u,x]/Sqrt[1+u^2],x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialQ[u,x]]
```

2: $\int (c + d\,x)^m \left(a + b\,\text{ArcSinh}[u]\right) dx$ when $m \neq -1 \land u$ is free of inverse functions

Derivation: Integration by parts

Rule: If $m \neq -1 \land u$ is free of inverse functions, then

$$\int (c + d\,x)^m \left(a + b\,\text{ArcSinh}[u]\right) dx \longrightarrow \frac{(c + d\,x)^{m+1} \left(a + b\,\text{ArcSinh}[u]\right)}{d\,(m + 1)} - \frac{b}{d\,(m + 1)} \int \frac{(c + d\,x)^{m+1}\,\partial_x u}{\sqrt{1 + u^2}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcSinh[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcSinh[u])/(d*(m+1)) -
  b/(d*(m+1))*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/Sqrt[1+u^2],x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && Not[FunctionOfExponentialQ[u,x]]
```

3: $\int v \left(a + b\,\text{ArcSinh}[u]\right) dx$ when u and $\int v\,dx$ are free of inverse functions

Derivation: Integration by parts

Rule: If u is free of inverse functions, let $w = \int v\,dx$, if w is free of inverse functions, then

$$\int v \left(a + b\,\text{ArcSinh}[u]\right) dx \longrightarrow w \left(a + b\,\text{ArcSinh}[u]\right) - b \int \frac{w\,\partial_x u}{\sqrt{1 + u^2}}\,dx$$

Program code:

```
Int[v_*(a_.+b_.*ArcSinh[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcSinh[u]),w,x] - b*Int[SimplifyIntegrand[w*D[u,x]/Sqrt[1+u^2],x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]]
```

8. $\int u\, e^{n\,\text{ArcSinh}[P_x]}\, dx$

1: $\int e^{n\,\text{ArcSinh}[P_x]}\, dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcSinh}[z]} == \left(z + \sqrt{1 + z^2}\right)^n$

Rule: If $n \in \mathbb{Z}$, then

$$\int e^{n\,\text{ArcSinh}[P_x]}\, dx \longrightarrow \int \left(P_x + \sqrt{1 + P_x{}^2}\right)^n dx$$

Program code:

```
Int[E^(n_.*ArcSinh[u_]), x_Symbol] :=
  Int[(u+Sqrt[1+u^2])^n,x] /;
IntegerQ[n] && PolyQ[u,x]
```

2: $\int x^m e^{n \text{ArcSinh}[P_x]} \, dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcSinh}[z]} == \left(z + \sqrt{1 + z^2}\right)^n$

Rule: If $n \in \mathbb{Z}$, then

$$\int x^m e^{n \text{ArcSinh}[P_x]} \, dx \longrightarrow \int x^m \left(P_x + \sqrt{1 + P_x{}^2}\right)^n \, dx$$

Program code:

```
Int[x_^m_.*E^(n_.*ArcSinh[u_]), x_Symbol] :=
  Int[x^m*(u+Sqrt[1+u^2])^n,x] /;
RationalQ[m] && IntegerQ[n] && PolyQ[u,x]
```